
Assembly Language:  
Function Calls 



Goals of this Lecture 

Help you learn: 
•  Function call problems 
•  IA-32 solutions 

•  Pertinent instructions and conventions 

Function Call Problems 
Calling and returning 

•  How does caller function jump to callee function? 
•  How does callee function jump back to the right place in caller 

function? 

Passing arguments 
•  How does caller function pass arguments to callee function? 

Storing local variables 
•  Where does callee function store its local variables? 

Returning a value 
•  How does callee function send return value back to caller function? 

Handling registers 
•  How do caller and callee functions use same registers without 

interference? 

Running Example 

void f(void) 
{  … 
   n = add3(3, 4, 5); 
   … 
} 

Caller 

Callee 
int add3(int a, int b, int c) 
{  int d; 
   d = a + b + c; 
   return d; 
} 



Agenda 
Calling and returning 

Passing arguments 

Storing local variables 

Returning a value 

Handling registers 

An example 

Problem 1: Calling and Returning 

How does caller jump to callee? 
•  I.e., Jump to the address of the callee’s first instruction 

How does the callee jump back to the right place in caller? 
•  I.e., Jump to the instruction immediately following the 

most-recently-executed call instruction 

1"

2"

void f(void) 
{  … 
   n = add3(3, 4, 5); 
   … 
} 

int add3(int a, int b, int c) 
{  int d; 
   d = a + b + c; 
   return d; 
} 



Attempted Solution: jmp Instruction 
Attempted solution: caller and callee use jmp instruction 

f: 

 … 

 jmp g  # Call g 

fReturnPoint: 

 … 

g: 

 … 

 jmp fReturnPoint  # Return 



Attempted Solution: jmp Instruction 
Problem: callee may be called by multiple callers 

f1: 

 … 

 jmp g  # Call g 

f1ReturnPoint: 

 … 

g: 

 … 

 jmp ???  # Return 

f2: 

 … 

 jmp g  # Call g 

f2ReturnPoint: 

 … 

Attempted Solution: Use Register 

f1: 

 movl $f1ReturnPoint, %eax 

 jmp g  # Call g 

f1ReturnPoint: 

 … 

f2: 

 movl $f2ReturnPoint, %eax 

 jmp g  # Call g 

f2ReturnPoint: 

 … 

g: 

 … 

 jmp *%eax  # Return 

Attempted solution: Store return address in register 

Special form of 
jmp instruction 

Attempted Solution: Use Register 

f: 

 movl $fReturnPoint, %eax 

 jmp g  # Call g 

fReturnPoint: 

 … 

g: 

 movl $gReturnPoint, %eax 

 jmp h  # Call h 

gReturnPoint: 

 … 

 jmp *%eax  # Return 

h: 

 … 

 jmp *%eax  # Return 

Problem if f() calls g(), and 
   g() calls h() 

Return address g() -> f() 
   is lost 

Problem: Cannot handle nested function calls 



IA-32 Solution: Use the Stack 
Observations: 

•  May need to store many return addresses 
•  The number of nested function calls is not known in advance 
•  A return address must be saved for as long as the invocation of 

this function is live, and discarded thereafter 
•  Stored return addresses are destroyed in 

reverse order of creation  
•  f() calls g() => return addr for g is stored 
•  g() calls h() => return addr for h is stored 
•  h() returns to g() => return addr for h is destroyed 
•  g() returns to f() => return addr for g is destroyed 

•  LIFO data structure (stack) is appropriate 

IA-32 solution:   
•  Use the STACK section of memory 
•  Via call and ret instructions 

EIP for f 

EIP for g 

EIP for h 

call and ret Instructions 

f: 

   … 

   call h 

   … 

   call g 

   … 

g: 

   … 

   call h 

   … 

   ret 

h: 

 … 

 ret 

ret instruction “knows” the return address 

1"
2"

3"
4"

5"

6"

Implementation of call 

Instruction Effective Operations 

pushl src subl $4, %esp 

movl src, (%esp) 

popl dest movl (%esp), dest 

addl $4, %esp 

ESP 

0 
ESP (stack pointer) register points 

to top of stack 

Implementation of call 

Instruction! Effective Operations!

pushl src subl $4, %esp 

movl src, (%esp) 

popl dest movl (%esp), dest 

addl $4, %esp 

call addr pushl %eip 
jmp addr 

ESP 
before 
call 

0 

Note: Can’t really access 
EIP directly, but this is 
implicitly what call is 
doing 

call instruction pushes return addr 
(old EIP) onto stack, then jumps 

EIP (instruction pointer) register 
points to next instruction to be 
executed 

Implementation of call 

Instruction Effective Operations 

pushl src subl $4, %esp 

movl src, (%esp) 

popl dest movl (%esp), dest 

addl $4, %esp 

call addr pushl %eip 
jmp addr ESP 

after 
call 

0 

Old EIP 

Implementation of ret 

Instruction Effective Operations 

pushl src subl $4, %esp 

movl src, (%esp) 

popl dest movl (%esp), dest 

addl $4, %esp 

call addr pushl %eip 
jmp addr 

ret popl %eip 
ESP 

before 
ret 

0 

Note: can’t really access 
EIP directly, but this is 
implicitly what ret is 
doing 

Old EIP 

ret instruction pops stack, thus 
placing return addr (old EIP) into EIP 

Implementation of ret 

Instruction Effective Operations 

pushl src subl $4, %esp 

movl src, (%esp) 

popl dest movl (%esp), dest 

addl $4, %esp 

call addr pushl %eip 
jmp addr 

ret popl %eip ESP 
after 
ret 

0 



Running Example 

Caller 

Callee 
f: 
   …    
   # Call the function 
   call add3 
   … 

add3: 
   …    
   ret 

Callee 



Agenda 
Calling and returning 

Passing arguments 

Storing local variables 

Returning a value 

Handling registers 

An example 

Problem 2: Passing Arguments 
Problem: How does caller pass arguments to callee? 

void f(void) 
{  … 
   n = add3(3, 4, 5); 
   … 
} 

int add3(int a, int b, int c) 
{  int d; 
   d = a + b + c; 
   return d; 
} 



IA-32 Solution: Use the Stack 
Observations (déjà vu): 

•  May need to store many arg sets 
•  The number of arg sets is not known in advance 
•  Arg set must be saved for as long as the invocation of this 

function is live, and discarded thereafter 
•  Stored arg sets are destroyed in 

reverse order of creation  
•  f() calls g() => arg set for g is created 
•  g() calls h() => arg set for h is created 
•  h() returns to g() => arg set for h is destroyed 
•  g() returns to f() => arg set for g is destroyed 

•  LIFO data structure (stack) is appropriate 

IA 32 solution:   
•  Use the STACK section of memory 

Arg set for f 

Arg set for g 

Arg set for h 

Passing Args on the Stack 

ESP 

0 Before executing call instruction… 

Passing Args on the Stack 

ESP 

0 Caller pushes args in reverse order 
•  Push Nth arg first 
•  Push 1st arg last 
•  So 1st arg is at top of  

the stack at the time of the call 

ArgN 

Arg1 

Arg… 

Passing Args on the Stack 

0 Caller executes call instruction 

ArgN 

Arg1 

Arg… 

ESP Old EIP 

Passing Args on the Stack 

ESP 

0 Callee accesses args relative to ESP 

Arg1 as 4(%esp) 

Arg2 as 8(%esp) 

… 

ArgN 

Arg1 

Arg… 

Old EIP 

Passing Args on the Stack 

ESP 

0 Callee executes ret instruction 

ArgN 

Arg1 

Arg… 

Passing Args on the Stack 

ESP 

0 Caller pops args from the stack 

Running Example 

f: 
   … 
   # Push arguments 
   pushl $5 
   pushl $4 
   pushl $3 

   # Call the function 
   call add3 

   # Pop arguments 
   addl $12, %esp 
   … 

add3: 
   … 
   # Use arguments 
   movl 4(%esp), %eax 
   addl 8(%esp), %eax 
   addl 12(%esp), %eax 
   …    
   ret 

Base Pointer Register: EBP 

ESP 

0 
Problem: 

•  As callee executes, ESP may 
change 
•  E.g., preparing to call another 

function 
•  Error-prone for callee to reference 

args as offsets relative to ESP 

Solution: 
•  Use EBP (base pointer) register 

•  EBP doesn’t change during 
callee’s execution 

•  Use EBP as fixed reference point to 
access args 

ArgN 

Arg1 

Arg… 

Old EIP 

EBP 

Passing Args on the Stack (v2) 

ESP 

0 Before executing call instruction… 

EBP 

Passing Args on the Stack (v2) 

ESP 

0 Caller pushes args in reverse order 
•  Push Nth arg first 
•  Push 1st arg last 
•  So 1st arg is at top of  

the stack at the time of the call 

ArgN 

Arg1 

Arg… 

EBP 

Passing Args on the Stack (v2) 

0 Caller executes call instruction 

ArgN 

Arg1 

Arg… 

ESP Old EIP 

EBP 

Need to save old value of EBP 
•  Before overwriting EBP register 

Callee executes “prolog” 
  pushl %ebp 

  movl %esp, %ebp 

Passing Args on the Stack (v2) 

ESP 

0 

ArgN 

Arg1 

Arg… 

Old EIP 

Old EBP 

EBP 

Callee executes “prolog” 
  pushl %ebp 

  movl %esp, %ebp 

Regardless of ESP, callee can 
reference Arg1 as 8(%ebp), 
Arg2 as 12(%ebp), etc. 

Passing Args on the Stack (v2) 

ESP 
EBP 

0 

ArgN 

Arg1 

Arg… 

Old EIP 

Old EBP 

Passing Args on the Stack (v2) 

EBP 

0 

ArgN 

Arg1 

Arg… 

Old EIP 

Old EBP 

Before returning, callee must 
restore ESP and EBP to their  
old values 

Callee executes “epilog” 
  movl %ebp, %esp 

  popl %ebp 

   

ESP 

Passing Args on the Stack (v2) 

ESP 
EBP 

0 

ArgN 

Arg1 

Arg… 

Old EIP 

Old EBP 

Callee executes “epilog” 
  movl %ebp, %esp 

  popl %ebp 

Passing Args on the Stack (v2) 

ESP 

0 

ArgN 

Arg1 

Arg… 

Old EIP 

Callee executes “epilog” 
  movl %ebp, %esp 

  popl %ebp 

   

EBP 

Passing Args on the Stack (v2) 

ESP 

0 

ArgN 

Arg1 

Arg… 

Callee executes ret instruction 

EBP 

Passing Args on the Stack (v2) 

ESP 

0 Caller pops args from the stack 

EBP 

Running Example 

f: 
   … 
   # Push arguments 
   pushl $5 
   pushl $4 
   pushl $3 

   # Call the function 
   call add3 

   # Pop arguments 
   addl $12, %esp 
   … 

add3: 
   pushl %ebp 
   movl %esp, %ebp 
   … 
   # Use arguments 
   movl 8(%ebp), %eax 
   addl 12(%ebp), %eax 
   addl 16(%ebp), %eax 
   … 
   movl %ebp, %esp 
   popl %ebp    
   ret 



Agenda 
Calling and returning 

Passing arguments 

Storing local variables 

Returning a value 

Handling registers 

An example 

Problem 3: Storing Local Variables 

Where does callee function store its local variables? 

void f(void) 
{  … 
   n = add3(3, 4, 5); 
   … 
} 

int add3(int a, int b, int c) 
{  int d; 
   d = a + b + c; 
   return d; 
} 



IA-32 Solution: Use the Stack 
Observations (déjà vu again!): 

•  May need to store many local var sets 
•  The number of local var sets is not known in advance 
•  Local var set must be saved for as long as the invocation of this 

function is live, and discarded thereafter 
•  Stored local var sets are destroyed in 

reverse order of creation  
•  f() calls g() => local vars set for g is created 
•  g() calls h() => local vars set for h is created 
•  h() returns to g() => local vars set for h is destroyed 
•  g() returns to f() => local vars set for g is destroyed 

•  LIFO data structure (stack) is appropriate 

IA 32 solution:   
•  Use the STACK section of memory 

Var set for f 

Var set for g 

Var set for h 



Storing Variables on the Stack 
Callee allocates space for its local 

variables on the stack 
•  Via pushl instructions 
•  Via subl $4,%esp instructions 

Example: allocate memory for 
two integers 
• subl  $4,%esp # int i; 
• pushl $5      # int j = 5; 

Callee references local variables as negative 
offsets relative to EBP 
• -4(%ebp)  # Access i 
• -8(%ebp)  # Access j 

EBP 

0 

ArgN 

Arg1 

Arg… 

Old EIP 

Old EBP 

ESP 

Var1 

Var2 

Running Example 

f: 
   … 
   # Push arguments 
   pushl $5 
   pushl $4 
   pushl $3 

   # Call the function 
   call add3 

   # Pop arguments 
   addl $12, %esp 
   … 

add3: 
   pushl %ebp 
   movl %esp, %ebp 

   # Allocate mem for local var 
   subl $4, %esp 

   # Use arguments 
   movl 8(%ebp), %eax 
   addl 12(%ebp), %eax 
   addl 16(%ebp), %eax 

   # Use local variable 
   movl %eax, -4(%ebp) 
   … 
   movl %ebp, %esp 
   popl %ebp    
   ret 



Agenda 
Calling and returning 

Passing arguments 

Storing local variables 

Returning a value 

Handling registers 

An example 

Problem 4: Return Values 
Problem: How does callee function send return value back to 

caller function? 

void f(void) 
{  … 
   n = add3(3, 4, 5); 
   … 
} 

int add3(int a, int b, int c) 
{  int d; 
   d = a + b + c; 
   return d; 
} 



IA-32 Solution: Use EAX 
In principle 

•  Store return value in stack frame of caller 

Or, for efficiency 
•  Known small size => store return value in register 
•  Other => store return value in stack 

IA-32 convention 
•  Integer or pointer: 

•  Store return value in EAX 
•  Floating-point number: 

•  Store return value in floating-point register 
•  (Beyond scope of COS 217) 

•  Structure: 
•  Store return value on stack 
•  (Beyond scope of COS 217) 

Running Example 
f: 
   … 
   # Push arguments 
   pushl $5 
   pushl $4 
   pushl $3 

   # Call the function 
   call add3 

   # Pop arguments 
   addl $12, %esp 

   # Use return value 
   movl %eax, n 
   … 

add3: 
   pushl %ebp 
   movl %esp, %ebp 

   # Allocate mem for local var 
   subl $4, %esp 

   # Use arguments 
   movl 8(%ebp), %eax 
   addl 12(%ebp), %eax 
   addl 16(%ebp), %eax 

   # Use local variable 
   movl %eax, -4(%ebp) 

   # Indicate return value    
   movl -4(%ebp), %eax 

   movl %ebp, %esp 
   popl %ebp    
   ret 



Agenda 
Calling and returning 

Passing arguments 

Storing local variables 

Returning a value 

Handling registers 

An example 

Problem 5: Handling Registers 

Observation: Registers are a finite resource 
•  In principle: Each function should have its own registers 
•  In reality:  All functions share same small set of registers 

Problem: How do caller and callee use same set of registers 
without interference? 
•  Callee may use register that the caller also is using 
•  When callee returns control to caller, old register contents may have 

been lost 
•  Caller function cannot continue where it left off 

IA-32 Solution: Register Conventions 

EBP 

0 

ArgN 

Arg1 
Arg… 

Old EIP 

Old EBP 

Callee-save registers: EBX, ESI, EDI 
•  If necessary… 

•  Callee saves to stack 
after prolog 

•  Callee restores from stack 
before epilog 

•  Caller can assume that values 
in EBX, ESI, EDI will not be 
changed by callee 

Caller-save registers: EAX, ECX, EDX 
•  If necessary… 

•  Caller saves to stack before call 
•  Caller restores from stack after call 

ESP 
Var1 

Var2 

Saved EAX, 
ECX,EDX 

Saved EBX, 
ESI,EDI 

Running Example 

f: 
   … 
   # Save EAX, ECX, EDX 
   pushl %eax 
   pushl %ecx 
   pushl %edx 

   # Push arguments 
   pushl $5 
   pushl $4 
   pushl $3 

   # Call the function 
   call add3 

   # Pop arguments 
   addl $12, %esp 

   # Use return value 
   movl %eax, n 

   # Restore EAX, ECX, EDX 
   popl %edx 
   popl %ecx 
   popl %eax 
   …  

Running Example 
add3: 
   pushl %ebp 
   movl %esp, %ebp 

   # Save EBX, ESI, EDI 
   pushl %ebx 
   pushl %esi 
   pushl %edi 

   # Allocate mem for local var 
   subl $4, %esp 

   # Use arguments 
   movl 8(%ebp), %eax 
   addl 12(%ebp), %eax 
   addl 16(%ebp), %eax 

   # Use local variable 
   movl %eax, -16(%ebp) 

   # Indicate return value    
   movl -4(%ebp), %eax 

   # Restore EBX, ESI, EDI 
   movl -12(%ebp), %edi 
   movl -8(%ebp), %esi 
   movl -4(%ebp), %ebx   

   movl %ebp, %esp 
   popl %ebp    
   ret 

Not necessary to save 
callee-save registers in 
this particular function 



Stack Frames 
Summary of IA-32 function handling: 

Stack has one stack frame per active function invocation 

ESP points to top (low memory) of current stack frame 

EBP points to bottom (high memory) of current stack frame 

Stack frame contains: 
•  Values of caller-save registers 
•  Arguments to be passed to callee function 
•  Return address (old EIP) 
•  Old EBP 
•  Values of callee-save registers 
•  Local variables 

Agenda 
Calling and returning 

Passing arguments 

Storing local variables 

Returning a value 

Handling registers 

An example 

56 



57 

int add3(int a, int b, int c) 
{  int d; 
   d = a + b + c; 
   return d; 
} 

void f(void) 
{  … 
   n = add3(3, 4, 5); 
   … 
} 

Trace of Running Example 

Trace of Running Example 1 

ESP 

EBP 

n = add3(3, 4, 5); 

High memory 

Low memory 

Trace of Running Example 2 

Old EAX 
Old ECX 
Old EDX ESP 

EBP 

# Save caller-save registers if necessary 
pushl %eax 
pushl %ecx 
pushl %edx 

n = add3(3, 4, 5); 

High memory 

Low memory 

Trace of Running Example 3 

Old EAX 
Old ECX 
Old EDX 

5 
4 
3 ESP 

EBP 

# Save caller-save registers if necessary 
pushl %eax 
pushl %ecx 
pushl %edx 
# Push arguments 
pushl $5 
pushl $4 
pushl $3 

n = add3(3, 4, 5); 

High memory 

Low memory 

Trace of Running Example 4 

# Save caller-save registers if necessary 
pushl %eax 
pushl %ecx 
pushl %edx 
# Push arguments 
pushl $5 
pushl $4 
pushl $3 
# Call add3 
call add3 Old EAX 

Old ECX 
Old EDX 

5 
4 
3 

Old EIP ESP 

EBP 

n = add3(3, 4, 5); 

High memory 

Low memory 

Trace of Running Example 5 

# Save old EBP 
pushl %ebp 

int add3(int a, int b, int c) 
{  int d; 
   d = a + b + c; 
   return d; 
} 

Old EAX 
Old ECX 
Old EDX 

5 
4 
3 

Old EIP 
Old EBP ESP 

EBP 

Prolog 

High memory 

Low memory 

Trace of Running Example 6 

# Save old EBP 
pushl %ebp 
# Change EBP 
movl %esp, %ebp 

int add3(int a, int b, int c) 
{  int d; 
   d = a + b + c; 
   return d; 
} 

EBP 

Old EAX 
Old ECX 
Old EDX 

5 
4 
3 

Old EIP 
Old EBP ESP 

Prolog 

High memory 

Low memory 

Trace of Running Example 7 

# Save old EBP 
pushl %ebp 
# Change EBP 
movl %esp, %ebp 
# Save callee-save registers if necessary 
pushl %ebx 
pushl %esi 
pushl %edi 

int add3(int a, int b, int c) 
{  int d; 
   d = a + b + c; 
   return d; 
} 

EBP 

Old EAX 
Old ECX 
Old EDX 

5 
4 
3 

Old EIP 
Old EBP 
Old EBX 
Old ESI 
Old EDI ESP 

Unnecessary here; add3 will not 
change the values in these registers 

High memory 

Low memory 

Trace of Running Example 8 

# Save old EBP 
pushl %ebp 
# Change EBP 
movl %esp, %ebp 
# Save caller-save registers if necessary 
pushl %ebx 
pushl %esi 
pushl %edi 
# Allocate space for local variable 
subl $4, %esp 

int add3(int a, int b, int c) 
{  int d; 
   d = a + b + c; 
   return d; 
} 

EBP 

Old EAX 
Old ECX 
Old EDX 

5 
4 
3 

Old EIP 
Old EBP 
Old EBX 
Old ESI 
Old EDI 

ESP 

High memory 

Low memory 

Trace of Running Example 9 

# Save old EBP 
pushl %ebp 
# Change EBP 
movl %esp, %ebp 
# Save caller-save registers if necessary 
pushl %ebx 
pushl %esi 
pushl %edi 
# Allocate space for local variable 
subl $4, %esp 
# Perform the addition 
movl 8(%ebp), %eax 
addl 12(%ebp), %eax 
addl 16(%ebp), %eax 
movl %eax, -16(%ebp) 

int add3(int a, int b, int c) 
{  int d; 
   d = a + b + c; 
   return d; 
} 

EBP 

Old EAX 
Old ECX 
Old EDX 

5 
4 
3 

Old EIP 
Old EBP 
Old EBX 
Old ESI 
Old EDI 

12 ESP 

High memory 

Low memory 

Access args as positive  
offsets relative to EBP 

Access local vars as neg  
offsets relative to EBP 

Trace of Running Example 10 

# Copy the return value to EAX 
movl -16(%ebp), %eax 
# Restore callee-save registers if necessary 
movl -12(%ebp), %edi 
movl -8(%ebp), %esi 
movl -4(%ebp), %ebx 

int add3(int a, int b, int c) 
{  int d; 
   d = a + b + c; 
   return d; 
} 

EBP 

Old EAX 
Old ECX 
Old EDX 

5 
4 
3 

Old EIP 
Old EBP 
Old EBX 
Old ESI 
Old EDI 

12 ESP 

High memory 

Low memory 

Trace of Running Example 11 

# Copy the return value to EAX 
movl -16(%ebp), %eax 
# Restore callee-save registers if necessary 
movl -12(%ebp), %edi 
movl -8(%ebp), %esi 
movl -4(%ebp), %ebx 
# Restore ESP 
movl %ebp, %esp 

int add3(int a, int b, int c) 
{  int d; 
   d = a + b + c; 
   return d; 
} 

EBP 

Old EAX 
Old ECX 
Old EDX 

5 
4 
3 

Old EIP 
Old EBP 
Old EBX 
Old ESI 
Old EDI 

12 

ESP 

Epilog 

High memory 

Low memory 

Trace of Running Example 12 

# Copy the return value to EAX 
movl -16(%ebp), %eax 
# Restore callee-save registers if necessary 
movl -12(%ebp), %edi 
movl -8(%ebp), %esi 
movl -4(%ebp), %ebx 
# Restore ESP 
movl %ebp, %esp 
# Restore EBP 
popl %ebp 

int add3(int a, int b, int c) 
{  int d; 
   d = a + b + c; 
   return d; 
} 

EBP 

Old EAX 
Old ECX 
Old EDX 

5 
4 
3 

Old EIP 
Old EBP 
Old EBX 
Old ESI 
Old EDI 

12 

ESP 

Epilog 

High memory 

Low memory 

Trace of Running Example 13 

# Copy the return value to EAX 
movl -16(%ebp), %eax 
# Restore callee-save registers if necessary 
movl -12(%ebp), %edi 
movl -8(%ebp), %esi 
movl -4(%ebp), %ebx 
# Restore ESP 
movl %ebp, %esp 
# Restore EBP 
popl %ebp 
# Return to calling function 
ret 

int add3(int a, int b, int c) 
{  int d; 
   d = a + b + c; 
   return d; 
} 

EBP 

Old EAX 
Old ECX 
Old EDX 

5 
4 
3 

Old EIP 
Old EBP 
Old EBX 
Old ESI 
Old EDI 

12 

ESP 

High memory 

Low memory 

Trace of Running Example 14 

# Save caller-save registers if necessary 
pushl %eax 
pushl %ecx 
pushl %edx 
# Push arguments 
pushl $5 
pushl $4 
pushl $3 
# Call add3 
call add3 
# Pop arguments 
addl $12, %esp 

EBP 

Old EAX 
Old ECX 
Old EDX 

5 
4 
3 

Old EIP 
Old EBP 
Old EBX 
Old ESI 
Old EDI 

12 

ESP 

n = add3(3, 4, 5); 

High memory 

Low memory 

Trace of Running Example 15 

# Save caller-save registers if necessary 
pushl %eax 
pushl %ecx 
pushl %edx 
# Push arguments 
pushl $5 
pushl $4 
pushl $3 
# Call add3 
call add3 
# Pop arguments 
addl %12, %esp 
# Use return value 
movl %eax, n 

Trace of Running Example 16 

# Save caller-save registers if necessary 
pushl %eax 
pushl %ecx 
pushl %edx 
# Push arguments 
pushl $5 
pushl $4 
pushl $3 
# Call add3 
call add3 
# Pop arguments 
addl %12, %esp 
# Use return value 
movl %eax, n 
# Restore caller-save registers if necessary 
popl %edx 
popl %ecx 
popl %eax 

n = add3(3, 4, 5); 

Trace of Running Example 17 

# Save caller-save registers if necessary 
pushl %eax 
pushl %ecx 
pushl %edx 
# Push arguments 
pushl $5 
pushl $4 
pushl $3 
# Call add3 
call add3 
# Pop arguments 
addl %12, %esp 
# Use return value 
movl %eax, n 
# Restore caller-save registers if necessary 
popl %edx 
popl %ecx 
popl %eax 
# Proceed! 
… 

n = add3(3, 4, 5); 

Summary 

Function calls in IA-32 assembly language 

Calling and returning 
• call instruction: push EIP onto stack and jump 
• ret instruction: pop from stack to EIP 

Passing arguments 
•  Caller pushes onto stack 
•  Callee accesses as positive offsets from EBP 
•  Caller pops from stack 

75 



Summary (cont.) 

Storing local variables 
•  Callee pushes onto stack 
•  Callee accesses as negative offsets from EBP 
•  Callee pops from stack 

Handling registers 
•  Caller saves and restores EAX, ECX, EDX if necessary 
•  Callee saves and restores EBX, ESI, EDI if necessary 

Returning values 
•  Callee places data of integer types and addresses in EAX 

